Problem:
Score : 300 points

Problem Statement

AtCoDeer the deer is going on a trip in a two-dimensional plane.
In his plan, he will depart from point (0, 0) at time 0, then for each i between 1 and N (inclusive), he will visit point (x_i,y_i) at time t_i.

If AtCoDeer is at point (x, y) at time t, he can be at one of the following points at time t+1: (x+1,y), (x-1,y), (x,y+1) and (x,y-1).
Note that he cannot stay at his place.
Determine whether he can carry out his plan.

Constraints

1 ≤ N ≤ 10^5

0 ≤ x_i ≤ 10^5

0 ≤ y_i ≤ 10^5

1 ≤ t_i ≤ 10^5

t_i < t_{i+1} (1 ≤ i ≤ N-1)

All input values are integers.

Input

Input is given from Standard Input in the following format:

N
t_1 x_1 y_1
t_2 x_2 y_2
:
t_N x_N y_N

Output

If AtCoDeer can carry out his plan, print Yes; if he cannot, print No.

Sample Input 1

2
3 1 2
6 1 1

Sample Output 1

Yes

For example, he can travel as follows: (0,0), (0,1), (1,1), (1,2), (1,1), (1,0), then (1,1).

Sample Input 2

1
2 100 100

Sample Output 2

No

It is impossible to be at (100,100) two seconds after being at (0,0).

Sample Input 3

2
5 1 1
100 1 1

Sample Output 3

No

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Control.Monad (replicateM)

reachable :: [(Int,Int,Int)] -> Bool
reachable ps = and $ zipWith reach ps (tail ps)

reach :: (Int, Int, Int) -> (Int, Int, Int) -> Bool
reach (t0,x0,y0) (t1,x1,y1) = accessible && sameParity
  where
    tick = t1 - t0
    shortest = abs (x1 - x0) + abs (y1 - y0)
    accessible = shortest <= tick
    sameParity = even (tick - shortest)

readTriple :: IO (Int, Int, Int)
readTriple = do
  [t,x,y] <- map read . words <$> getLine
  return (t,x,y)

main :: IO ()
main = do
  n <- readLn
  ps <- replicateM n readTriple
  putStrLn $ if reachable ((0,0,0):ps) then "Yes" else "No"